// File: src/Crowdfundoor.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.20;

//
//                            _  __                 _
//                           | |/ _|               | |
//   ___ _ __ _____      ____| | |_ _   _ _ __   __| | ___   ___  _ __
//  / __| '__/ _ \ \ /\ / / _` |  _| | | | '_ \ / _` |/ _ \ / _ \| '__|
// | (__| | | (_) \ V  V / (_| | | | |_| | | | | (_| | (_) | (_) | |
//  \___|_|  \___/ \_/\_/ \__,_|_|  \__,_|_| |_|\__,_|\___/ \___/|_|
//
//                                                 by netdragonx.eth
//
import "@openzeppelin/contracts/token/ERC721/IERC721.sol";

struct Campaign {
    uint256 amount;
    uint256 tokenId;
    address tokenAddress;
    address recipient;
    bool isAccepted;
    mapping(address => uint256) donations;
}

contract Crowdfundoor {
    mapping(uint256 => Campaign) public campaigns;
    uint256 public nextCampaignId;

    error AlreadyOwned();
    error AmountLessThanMinimum();
    error DonationRequired();
    error InvalidRecipient();
    error NoDonationToWithdraw();
    error NotTokenOwner();
    error TransferFailed();
    error CampaignAlreadyAccepted();

    event Start(uint256 indexed campaignId, address indexed tokenAddress, uint256 indexed tokenId, address recipient);
    event Donation(uint256 indexed campaignId, address indexed donor, uint256 amount);
    event Withdrawal(uint256 indexed campaignId, address indexed donor, uint256 amount);
    event Accepted(uint256 indexed campaignId);

    function startCampaign(address tokenAddress, uint256 tokenId, address recipient) external returns (uint256) {
        if (IERC721(tokenAddress).ownerOf(tokenId) == recipient) {
            revert AlreadyOwned();
        }

        uint256 campaignId = nextCampaignId;

        Campaign storage newCampaign = campaigns[campaignId];
        newCampaign.tokenAddress = tokenAddress;
        newCampaign.tokenId = tokenId;
        newCampaign.recipient = recipient;

        unchecked {
            nextCampaignId++;
        }

        emit Start(campaignId, tokenAddress, tokenId, recipient);

        return campaignId;
    }

    function donate(uint256 campaignId) external payable {
        if (msg.value == 0) {
            revert DonationRequired();
        }

        Campaign storage campaign = campaigns[campaignId];

        if (campaign.isAccepted) {
            revert CampaignAlreadyAccepted();
        }

        campaign.amount += msg.value;
        campaign.donations[msg.sender] += msg.value;

        emit Donation(campaignId, msg.sender, msg.value);
    }

    function accept(uint256 campaignId, uint256 minimumAmount) external {
        Campaign storage campaign = campaigns[campaignId];

        if (campaign.amount < minimumAmount) {
            revert AmountLessThanMinimum();
        }

        if (IERC721(campaign.tokenAddress).ownerOf(campaign.tokenId) != msg.sender) {
            revert NotTokenOwner();
        }

        uint256 amount = campaign.amount;
        campaign.amount = 0;
        campaign.isAccepted = true;

        IERC721(campaign.tokenAddress).transferFrom(msg.sender, campaign.recipient, campaign.tokenId);

        (bool success,) = payable(msg.sender).call{value: amount}("");
        if (!success) revert TransferFailed();

        emit Accepted(campaignId);
    }

    function withdraw(uint256 campaignId) external {
        Campaign storage campaign = campaigns[campaignId];

        if (campaign.isAccepted) {
            revert CampaignAlreadyAccepted();
        }

        uint256 donation = campaign.donations[msg.sender];
        if (donation == 0) {
            revert NoDonationToWithdraw();
        }

        campaign.donations[msg.sender] = 0;
        campaign.amount -= donation;

        (bool success,) = payable(msg.sender).call{value: donation}("");
        if (!success) revert TransferFailed();

        emit Withdrawal(campaignId, msg.sender, donation);
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes calldata data) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: lib/openzeppelin-contracts/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}